# Additional import path used to resolve QML modules in Creator's code model
QML_IMPORT_PATH =

# Avoid auto screen rotation
##DEFINES += ORIENTATIONLOCK

# Needs to be defined for Symbian
##DEFINES += NETWORKACCESS

symbian:TARGET.UID3 = 0xE1E784F6

# Smart Installer package's UID
# This UID is from the protected range 
# and therefore the package will fail to install if self-signed
# By default qmake uses the unprotected range value if unprotected UID is defined for the application
# and 0x2002CCCF value if protected UID is given to the application
##symbian:DEPLOYMENT.installer_header = 0x2002CCCF

# Define QMLJSDEBUGGER to allow debugging of QML in debug builds
# (This might significantly increase build time)
##DEFINES += QMLJSDEBUGGER

# If your application uses the Qt Mobility libraries, uncomment the following
# lines and add the respective components to the MOBILITY variable.
# NOTE : multimedia is recently moved from QT to MOBILITY.
CONFIG += mobility link_pkgconfig qdeclarative-boostable
MOBILITY += sensors systeminfo multimedia
QT += opengl dbus

# Add extra libraries here if needed.
##LIBS += -lpulse -lglib-2.0 -lgstreamer-0.10

PKGCONFIG += \
    libpulse \
    gstreamer-0.10 \
    qdeclarative-boostable

# The .cpp file which was generated for your project. Feel free to hack it.
SOURCES += main.cpp \
    gamewidget.cpp \
    gamesettings.cpp \
    gamewidget_GL.cpp \
    gamewidget_GLes2.cpp \
    gamewidget_graphics.cpp \
    spacecraft.cpp \
    gameobject.cpp \
    asteroid.cpp \
    gameappviewer.cpp \
    blackhole.cpp \
    ufo.cpp \
    shaderprograms.cpp \
    frameinfo.cpp \
    gameeffect.cpp \
    soundengine.cpp \
    wavfile.cpp \
    collectible.cpp \
    timermanager.cpp \
    gametimer.cpp \
    visualeffect.cpp \
    gamelevel.cpp \
    hyperjumpnavigator.cpp \
    rotationsensor.cpp \
    accelometersensor.cpp \
    gameengine_testing.cpp \
    visualeffectdata.cpp \
    soundengineqaudio.cpp \
    soundenginepulseaudio.cpp \
    pulseaudiooutput.cpp \
    gameengine.cpp \
    rotatingobject.cpp \
    musicplayer.cpp \
    particledata.cpp \
    comet.cpp \
    missile.cpp \
    highscoremodel.cpp \
    meteorshower.cpp

# Please do not modify the following two lines. Required for deployment.
include(qmlapplicationviewer/qmlapplicationviewer.pri)
qtcAddDeployment()

HEADERS += \
    gamewidget.h \
    config.h \
    gamesettings.h \
    spacecraft.h \
    gameobject.h \
    asteroid.h \
    gameappviewer.h \
    blackhole.h \
    ufo.h \
    shaderprograms.h \
    frameinfo.h \
    gameeffect.h \
    soundengine.h \
    wavfile.h \
    collectible.h \
    timermanager.h \
    gametimer.h \
    visualeffect.h \
    gamelevel.h \
    hyperjumpnavigator.h \
    rotationsensor.h \
    accelometersensor.h \
    visualeffectdata.h \
    soundengineqaudio.h \
    soundenginepulseaudio.h \
    pulseaudiooutput.h \
    gameengine.h \
    rotatingobject.h \
    musicplayer.h \
    particledata.h \
    comet.h \
    missile.h \
    highscoremodel.h \
    meteorshower.h \
    defines.h

TRANSLATIONS = ts/smashminer_en.ts

RESOURCES += \
    translations.qrc \
    smashminer.qrc

# To build qm translation files
!isEmpty(TRANSLATIONS) {
    isEmpty(QMAKE_LRELEASE) {
        win32:QMAKE_LRELEASE = $$[QT_INSTALL_BINS]\lrelease.exe
        else:QMAKE_LRELEASE = $$[QT_INSTALL_BINS]/lrelease
    }

isEmpty(TS_DIR):TS_DIR = Translations
    TSQM.name = lrelease ${QMAKE_FILE_IN}
    TSQM.input = TRANSLATIONS
    TSQM.output = $$TS_DIR/${QMAKE_FILE_BASE}.qm
    TSQM.commands = $$QMAKE_LRELEASE ${QMAKE_FILE_IN}
    TSQM.CONFIG = no_link
    QMAKE_EXTRA_COMPILERS += TSQM
    PRE_TARGETDEPS += compiler_TSQM_make_all
} else:message(No translation files in project)

#unix:!symbian {
unix {

    target.path = /opt/smashminer/bin

    desktopfile.files = $${TARGET}.desktop
    maemo5 {
        desktopfile.path = /usr/share/applications/hildon
    } else {
        desktopfile.path = /usr/share/applications
    }

    iconfile.files += $${TARGET}.png
    maemo5 {
      iconfile.path = /usr/share/icons/hicolor/64x64/apps
    } else {
      iconfile.path = /usr/share/icons/hicolor/64x64/apps
    }

    graphics.files += \
        qml/smashminer/pics/1024x512_01.jpg \
        qml/smashminer/pics/1024x512_02.jpg \
        qml/smashminer/pics/1024x512_03.jpg \
        qml/smashminer/pics/1024x512_04.jpg \
        qml/smashminer/pics/1024x512_05.jpg \
        qml/smashminer/pics/1024x512_06.jpg \
        qml/smashminer/pics/1024x512_07.jpg \
        qml/smashminer/pics/1024x512_08.jpg \
        qml/smashminer/pics/1024x512_09.jpg \
        qml/smashminer/pics/1024x512_10.jpg \
        qml/smashminer/pics/1024x512_11.jpg \
        qml/smashminer/pics/1024x512_12.jpg \
        qml/smashminer/pics/1024x512_13.jpg \
        qml/smashminer/pics/1024x512_14.jpg \
        qml/smashminer/pics/1024x512_15.jpg \
        qml/smashminer/pics/1024x512_16.jpg \
        qml/smashminer/pics/1024x512_17.jpg \
        qml/smashminer/pics/1024x512_18.jpg \
        qml/smashminer/pics/1024x512_19.jpg \
        qml/smashminer/pics/1024x512_20.jpg
    graphics.path = /opt/smashminer

    music.files += \
        sounds/oikophobia.ogg \
        sounds/oikophobia.mp3
    music.path = /opt/smashminer

    ########################################
    # MeeGo 1.2 Harmattan
    # --------------------------------------
    # Lets classify as gaming
    # http://wiki.meego.com/images/Meego-policy-framework-developer-guide.pdf
    policyfile.files += smashminer.conf
    policyfile.path = /usr/share/policy/etc/syspart.conf.d/

    splashscreen.files += splashscreen.png
    splashscreen.path = /opt/smashminer

    CONFIG += qdeclarative-boostable

    INSTALLS += target desktopfile iconfile graphics music policyfile splashscreen

}

OTHER_FILES += \
    qml/smashminer/ArrowButton.qml \
    qml/smashminer/Switch.qml \
    qml/smashminer/Slider.qml \
    qml/smashminer/Settings.qml \
    qml/smashminer/main.qml \
    qml/smashminer/HighScores.qml \
    qml/smashminer/Help.qml \
    qml/smashminer/Controls.qml \
    qml/smashminer/ControlsPage1.qml \
    qml/smashminer/Button.qml \
    qml/smashminer/ControlsPage2.qml \
    qml/smashminer/ImageCell.qml \
    qml/smashminer/TextCell.qml \
    qml/smashminer/ViewHeader.qml \
    qml/smashminer/ViewFooter.qml \
    qml/smashminer/HelpText.qml \
    qml/smashminer/HelpPage1.qml \
    qml/smashminer/HelpPage2.qml \
    qml/smashminer/HelpPage3.qml \
    qml/smashminer/HelpPage4.qml \
    qml/smashminer/HelpPage5.qml \
    qtc_packaging/debian_harmattan/rules \
    qtc_packaging/debian_harmattan/README \
    qtc_packaging/debian_harmattan/copyright \
    qtc_packaging/debian_harmattan/control \
    qtc_packaging/debian_harmattan/compat \
    qtc_packaging/debian_harmattan/changelog \
    qtc_packaging/debian_harmattan/manifest.aegis \
    smashminer.conf \
    splashscreen.png

